VICTOR T.TOTH

1 http://www.vttoth.com/CMS/projects/61-the-maxima-computer-algebra-system

Example of using Maxima for derivation of Schwartzschild metric from Hilbert Variational principle.

Below L is Lambda-term.

Slightly corrected by S.Blinnikov on 21st October 2012 (Novosibirsk): Friedmann lines are removed.

```
(%i1) if get('ctensor,'version)=false then load(ctensor);
```

(%o1) */usr/share/maxima/*5.24.0*/share/tensor/ctensor.mac*

```
(%i2) if get('itensor,'version)=false then load(itensor);
```

(%o2) */usr/share/maxima/*5.24.0*/share/tensor/itensor.lisp*

```
(%i3) remsym(g,2,0);
      remsym(g,0,2);
      remsym(gg,2,0);
      remsym(gg,0,2);
      remcomps(gg);
      imetric(gg);
```

No symmetries have been declared for this tensor.

(%o3) *done*

No symmetries have been declared for this tensor.

(%o4) *done*

No symmetries have been declared for this tensor.

(%o5) *done*

No symmetries have been declared for this tensor.

(%o6) *done*

(%o7) *done*

(%o8) *done*

```
(%i9) icurvature([a,b,c],[e])*gg([d,e],[])$
```

```
(%i10) contract(rename(expand(%)))$

(%i11) %,ichr2$

(%i12) contract(rename(expand(%)))$

(%i13) canform(%)$

(%i14) contract(rename(expand(%)))$

(%i15) components(gg([a,b],[]),kdels([a,b],[u,v])*g([u,v],[])/2);
(%o15) done

(%i16) components(gg([],[a,b]),kdels([u,v],[a,b])*g([],[u,v])/2);
(%o16) done

(%i17) %th(4),gg$

(%i18) contract(rename(expand(%)))$

(%i19) contract(canform(%))$

(%i20) imetric(g);
(%o20) done

(%i21) contract(rename(expand(%th(2))))$

(%i22) remcomps(R);
(%o22) done

(%i23)  components(R([a,b,c,d],[]),%th(2));
(%o23) done

(%i24) g([],[a,b])*R([a,b,c,d])*g([],[c,d])$

(%i25) contract(rename(canform(%)))$

(%i26) contract(rename(canform(%)))$
```

```
(%i27) components(R([],[]),%);
```

(%o27) *done*

```
(%i28) decsym(g,2,0,[sym(all)],[]);
```

(%o28) *done*

```
(%i29) decsym(g,0,2,[],[sym(all)]);
```

(%o29) *done*

```
(%i30) ishow(1/(16*%pi*G)*((2*L+'R([],[]))*sqrt(-determinant(g)))$
```

$$(\%t30)\quad \frac{\sqrt{-\mathrm{determinant}\,(g)}\;(2\,L+R)}{16\,\pi\,G}$$

```
(%i31) L0:%,R$
```

```
(%i32) canform(contract(canform(rename(contract(expand(diff(L0,g([],[m,n]))-
       idiff(diff(L0,g([],[m,n],k)),k)+idiff(rename(idiff(contract(
       diff(L0,g([],[m,n],k,l))),k),1000),l)))))))$
```

```
(%i33) ishow(e([m,n],[])=canform(%*16*%pi/sqrt(-determinant(g))))$
```

$$(\%t33)\quad e_{mn}=-\frac{g_{mn}\,L}{G}+\frac{g_{,\%3}^{\%1\%2}\,g^{\%3\%5}\,g_{\%1\%2}\,ichr2_{\%4\%5}^{\%4}\,g_{mn}}{2\,G}+\frac{g_{,\%3}^{\%1\%2}\,g^{\%3\%6}\,g_{,\%6}^{\%4\%5}\,g_{\%1\%2}\,g_{\%4\%5}\,g_{mn}}{8\,G}+$$

$$\frac{g^{\%2\%4}\,ichr2_{\%1\%2}^{\%1}\,ichr2_{\%3\%4}^{\%3}\,g_{mn}}{G}+\frac{g_{,\%2}^{\%1\%2}\,g_{,\%1}^{\%3\%4}\,g_{\%3\%4}\,g_{mn}}{8\,G}-\frac{3\,g_{,\%3}^{\%1\%2}\,g^{\%3\%6}\,g_{,\%6}^{\%4\%5}\,g_{\%1\%4}\,g_{\%2\%5}\,g_{mn}}{16\,G}+$$

$$\frac{5\,g_{,\%3}^{\%1\%2}\,g^{\%3\%6}\,g_{,\%6}^{\%4\%5}\,g_{\%1\%5}\,g_{\%2\%4}\,g_{mn}}{16\,G}-\frac{g_{,\%3}^{\%1\%2}\,g_{,\%1}^{\%3\%4}\,g_{\%2\%4}\,g_{mn}}{4\,G}+\frac{g_{,\%1}^{\%1\%2}\,ichr2_{\%2\%3}^{\%3}\,g_{mn}}{2\,G}+$$

$$\frac{g_{,\%2}^{\%1\%2}\,ichr2_{\%1\%3}^{\%3}\,g_{mn}}{2\,G}+\frac{g^{\%2\%3}\,ichr2_{\%1\%2,\%3}^{\%1}\,g_{mn}}{G}-\frac{g_{,\%3}^{\%1\%2}\,g_{,\%4}^{\%3\%4}\,g_{\%1\%2}\,g_{mn}}{8\,G}+$$

$$\frac{g_{,\%1\%2}^{\%1\%2}\,g_{mn}}{2\,G}+\frac{g_{,\%3}^{\%1\%2}\,g^{\%3\%5}\,g_{\%1n}\,g_{\%2m}\,ichr2_{\%4\%5}^{\%4}}{2\,G}-\frac{g_{,n}^{\%1\%2}\,g_{\%1\%2}\,ichr2_{\%3m}^{\%3}}{2\,G}+\frac{g_{,n}^{\%1\%2}\,g_{,\%1}^{\%3\%4}\,g_{\%2\%4}\,g_{\%3m}}{4\,G}+$$

$$\frac{g_{,n}^{\%1\%2}\,g_{,\%2}^{\%3\%4}\,g_{\%1\%4}\,g_{\%3m}}{4\,G}-\frac{g_{,\%6}^{\%1\%5}\,g_{,\%4}^{\%2\%3}\,g^{\%4\%6}\,g_{\%1n}\,g_{\%2m}\,g_{\%3\%5}}{2\,G}-\frac{g_{,n}^{\%1\%2}\,g_{,m}^{\%3\%4}\,g_{\%1\%2}\,g_{\%3\%4}}{4\,G}-$$

$$\frac{ichr2_{\%1n}^{\%1}\,ichr2_{\%2m}^{\%2}}{G}+\frac{9\,g_{,\%3}^{\%1\%2}\,g_{,\%4}^{\%3\%4}\,g_{\%1n}\,g_{\%2m}}{16\,G}-\frac{g_{,\%4}^{\%1\%2}\,g_{,\%3}^{\%3\%4}\,g_{\%1n}\,g_{\%2m}}{16\,G}+\frac{g_{,\%3\%4}^{\%1\%2}\,g^{\%3\%4}\,g_{\%1n}\,g_{\%2m}}{2\,G}-$$

$$\frac{g_{,\%3}^{\%1\%4}\,g_{,\%4}^{\%2\%3}\,g_{\%1n}\,g_{\%2m}}{2\,G}-\frac{g_{,n}^{\%1\%2}\,ichr2_{\%1\%3}^{\%3}\,g_{\%2m}}{4\,G}-fracg_{,\%1n}^{\%1\%2}\,g_{\%2m}4\,G+\frac{g_{,\%3}^{\%1\%4}\,g_{,m}^{\%2\%3}\,g_{\%1n}\,g_{\%2\%4}}{2\,G}+$$

$$\frac{3\,g_{,n}^{\%1\%2}\,g_{,m}^{\%3\%4}\,g_{\%1\%3}\,g_{\%2\%4}}{8\,G}-\frac{g_{,m}^{\%1\%2}\,g_{\%1n}\,ichr2_{\%2\%3}^{\%3}}{2\,G}-\frac{g_{,n}^{\%1\%2}\,g_{\%1m}\,ichr2_{\%2\%3}^{\%3}}{4\,G}-$$

$$\frac{g_{,m}^{\%2\%3}\,ichr2_{\%1n}^{\%1}\,g_{\%2\%3}}{2\,G}-\frac{5\,g_{,n}^{\%1\%2}\,g_{,m}^{\%3\%4}\,g_{\%1\%4}\,g_{\%2\%3}}{8\,G}-\frac{g_{,\%2m}^{\%1\%2}\,g_{\%1n}}{2\,G}-\frac{ichr2_{\%1m,n}^{\%1}}{G}-$$

$$\frac{g_{,\%2n}^{\%1\%2}\, g_{\%1m}}{4\,G}$$

```
(%i34) EQ:ic_convert(%)$
```

```
(%i35) ct_coords:[t,r,u,v];
```

(%o35) $[t, r, u, v]$

```
(%i36) lg:ident(4);
```

(%o36) $$\begin{pmatrix}1 & 0 & 0 & 0\\ 0 & 1 & 0 & 0\\ 0 & 0 & 1 & 0\\ 0 & 0 & 0 & 1\end{pmatrix}$$

```
(%i37) lg[1,1]:B;
       lg[2,2]:-A;
       lg[3,3]:-r^2;
       lg[4,4]:-r^2*sin(u)^2;
```

(%o37) B

(%o38) $-A$

(%o39) $-r^2$

(%o40) $-r^2 \sin(u)^2$

```
(%i41) kill(dependencies);
```

(%o41) *done*

```
(%i42) dependencies(A(r),B(r));
```

(%o42) $[\mathrm{A}(r), \mathrm{B}(r)]$

```
(%i43) cmetric();
```

(%o43) *done*

```
(%i44) christof(false);
```

(%o44) *done*

```
(%i45) e:zeromatrix(4,4);
```

(%o45) $$\begin{pmatrix}0 & 0 & 0 & 0\\ 0 & 0 & 0 & 0\\ 0 & 0 & 0 & 0\\ 0 & 0 & 0 & 0\end{pmatrix}$$

(%i46) `ev(EQ);`

(%o46) *done*

(%i47) `E:expand(radcan(ug.e));`

(%o47) $$\begin{pmatrix} -\frac{L}{G}+\frac{\frac{d}{d\,r}\,A}{r\,A^2\,G}-\frac{1}{r^2\,A\,G}+\frac{1}{r^2\,G} & 0 & 0 \\ 0 & -\frac{L}{G}-\frac{\frac{d}{d\,r}\,B}{r\,A\,B\,G}-\frac{1}{r^2\,A\,G}+\frac{1}{r^2\,G} & 0 \\ 0 & 0 & -\frac{L}{G}-\frac{\frac{d^2}{d\,r^2}\,B}{2\,A\,B\,G}+\frac{\left(\frac{d}{d\,r}\,B\right)^2}{4\,A\,B^2\,G}+\frac{\left(\frac{d}{d\,r}\,A\right)\left(\frac{d}{d\,r}\,B\right)}{4\,A^2\,B\,G} \\ 0 & 0 & 0 \end{pmatrix}$$

(%i48) `exp:findde(E,2);`

(%o48) $$[r^2\,A^2\,L-r\left(\frac{d}{d\,r}\,A\right)-A^2+A, r^2\,A\,B\,L+r\left(\frac{d}{d\,r}\,B\right)-A\,B+B, 4\,r\,A^2\,B^2\,L+ 2\,r\,A\,B\left(\frac{d^2}{d\,r^2}\,B\right)-r\,A\left(\frac{d}{d\,r}\,B\right)^2-r\left(\frac{d}{d\,r}\,A\right)B\left(\frac{d}{d\,r}\,B\right)+2\,A\,B\left(\frac{d}{d\,r}\,B\right)- 2\left(\frac{d}{d\,r}\,A\right)B^2]$$

(%i49) `solve(ode2(exp[1],A,r),A);`

(%o49) $$[A=-\frac{3\,r}{r^3\,L-3\,r-3\,\%c}]$$

(%i50) `%,%c=-2*M;`

(%o50) $$[A=-\frac{3\,r}{6\,M+r^3\,L-3\,r}]$$

(%i51) `a:%[1],%c=-2*M;`

(%o51) $$A=-\frac{3\,r}{6\,M+r^3\,L-3\,r}$$

(%i52) `ode2(ev(exp[2],a),B,r);`

(%o52) $$B=\frac{\%c\,\left(6\,M+r^3\,L-3\,r\right)}{r}$$

(%i53) `b:ev(%,%c=rhs(solve(rhs(%)*rhs(a)=1,%c)[1]));`

(%o53) $$B=-\frac{6\,M+r^3\,L-3\,r}{3\,r}$$

(%i54) `factor(ev(ev(exp[3],a,b),diff));`

(%o54) 0

```
(%i55) lg:ev(lg,a,b),L=0$

(%i56) ug:invert(lg)$

(%i57) block([title: "Schwarzschild Potential for Mass M=2",M:2.],
       wxplot3d([r*cos(th),r*sin(th),1-ug[1,1]],[r,5.,50.],[th,-%pi,%pi],
       ['grid,20,30],['z,-2,0],[psfile],['legend,title]));
```

(%t57)